

MAPLE Course

Numerical investigation is a key component in the study of mathematical models. In MATH 111 we will use a package called Maple to gain insight into the behaviour of mathematical models.

This worksheet covers basic Maple commands. You should work your way through this sheet, entering commands into Maple that are typed in this font; answering questions and making notes where appropriate. The purpose of this worksheet is to show you some of the basic operations that Maple can do. Don't worry about trying to learn the commands and syntax.

Note. Maple is case *sensitive*.

- You should finish this worksheet before the lab in week 2 — you will be given a new worksheet in week 2.
- You are *expected* to bring this worksheet to your lab in week 3.
 - You *will* be asked in week 3 to *demonstrate* that you have worked through the worksheet.
 - Students who demonstrate that they have finished this worksheet will be rewarded, those who can not will be penalised.
- You may find it useful to bring this worksheet to labs from weeks 3 on.

1 BasicS

1.1 Basic basics

Operations

+	addition	<	less than
-	subtraction	<=	less or equal
*	multiplication	>	greater than
/	division	>=	greater or equal
^	exponentiation	<>	not equal

Special Constants

e,E	not a reserved name, use exp(1) for 'e'
I	complex symbol, $I^2 = -1$
infinity	mathematical infinity ∞
Pi	constant π , evalf(Pi) is approximately 3.14159265...

Functions

Built-in functions in MAPLE include

sin, cos, tan, cosec, sec, cot, arcsin, arccos, arctan, exp, log, sqrt, abs, cosh, sinh, tanh, ...

There are many other built-in functions that are not mentioned here. Refer to the HELP menu in MAPLE.

Some mathematical operations on numbers, expressions and functions

We can use MAPLE as a calculator to perform basic calculations.

Question 1 Try the following inputs and record the outputs:

Numbers	Algebraic
1+2;	a+b;
1-2;	a-b;
5*6;	a*b;
20/3;	a/b;

Note the MAPLE response to division of numbers. MAPLE accepts 20/3; in rational form (just as we would). To evaluate 20/3, we need to enter

```
evalf ( 20 / 3 );
```

Question 2 What happens when we enter

1. 20 / 3.0 ;
2. 20.0 / 3 ;

Note that the returned values have 10 digits. This is the default in MAPLE. We can check the number of digits to be returned by entering

```
Digits;
```

Whenever we start a new MAPLE session, Digits will be initially be set to 10. We can vary the digits to be displayed by entering

```
Digits := n;      where n is any natural number.
```

For example, if we want 20 digits, we enter

```
Digits := 20;
```

```
evalf ( 20 / 3 );
```

If we want 5 digits, we enter

```
Digits := 5;
```

```
evalf ( 20 / 3 );
```

Before carrying out set Digits back to 10.

Question 3 What command do you use to set Digits to 10?

Numbers	Algebraic
(-1) ^3;	a ^ b;
2 ^ (1 / 2);	a ^ (1 / 2);
sqrt (2);	sqrt (a);
2 ^ (1 / 3);	

The last expression is an irrational number. To evaluate it, we need to enter

```
evalf ( 2 ^ ( 1 / 3 ) );
```

Question 4 What happens when we enter

```
sqrt ( 2.0 );
```

Question 5 Evaluate the following inputs. Explain your answers.

- (1) cos(pi);
- (2) cos(Pi);
- (3) Cos(pi);
- (4) Cos(Pi);
- (5) evalf(cos(pi));
- (6) evalf(cos(Pi));
- (7) evalf(Cos(pi));
- (8) evalf(Cos(Pi));

Question 6 Evaluate the following inputs:

```
1. cos ( Pi / 4 );
2. cos ( Pi / 12 );
```

To get the numerical value 0.9659258263 for $\cos(\frac{\pi}{12})$, we need to enter

```
evalf ( cos ( Pi / 12 ) );
```

Question 7 *What happens when we enter*

```
cos ( Pi / 12.0 );
```

1.2 Simplifying

MAPLE always computes the result to the five arithmetic operations $x + y$, $x - y$, $x \times y$, x/y and x^n if n is an integer and x and y are numbers. For example, you cannot prevent MAPLE from simplifying $2 + 3$ to 5 in output. If the operands are floating-point numbers, MAPLE performs the operations in the floating-point environment.

Question 8 *Evaluate the following commands.*

```
2+3;      6/4;      1.2 / 7;      (2 + I)/(2 - 2 * I );
```

Question 9 *MAPLE automatically simplifies the following expressions for any symbol x or arbitrary expression. To what does Maple simplify them?*

```
x-x;  x+x;  x+0;  x*x;
x/x;  x*1;  x^0;  x^1;
```

Note the following exceptions. The commands

```
infinity - infinity;      and      infinity / infinity;
```

return an undefined comment, whilst the input

```
0 / 0;
```

returns an error message.

Question 10 *What error message does the last input return?*

Question 11 *Enter the following Maple commands. After each command record the Maple output.*

```
4 ^ ( 1 / 2 ) + 3;
simplify ( % );
```

See ? simplify for more information.

Note the use of the **ditto** operator. The sequence of expressions assigned to the three ditto operators %, %% and %%% are the last three non-null results generated in the MAPLE session independent of where they are located in the session (this means the last three inputs in your worksheet may not be the last three entries).

Try the following:

```
x ^ 2 - y ^ 2;
```

To get MAPLE to factorise this expression, we enter

```
factor (%);      (using the ditto operator introduced above)
```

Now try

```
factor ( x ^ 2 - 2 );
```

We see that MAPLE leaves the expression unchanged. We can force MAPLE to factorise some expression this type by giving it a field extension over which to factor. Try

```
factor ( x ^ 2 - 2 , sqrt ( 2 ) );
```

See ?factor for more information.

1.3 Complex numerical constants

(If you didn't cover complex numbers in high school mathematics, don't worry. We won't be using complex numbers in MATH 111 and you *will* study them in other first-year mathematics subjects).

Assume we have $z = 2 + 3i$. We enter

```
z := 2 + 3 * I;
```

To select the real and imaginary parts, we enter

```
Re ( z );      and      Im ( z );
```

MAPLE can evaluate elementary functions and many special functions over the complex numbers. MAPLE evaluates the result automatically if **a** and **b** are numeric constants and one of **a** or **b** is a decimal number. Record the output from the following commands.

```
exp ( 2 + 3 * I );      exp ( 2 + 3.0 * I );
```

If the arguments are not complex floating-point constants, you can expand the expression in some cases to the form $\mathbf{a} + \mathbf{bi}$, where **a** and **b** are real, by using the evalc command. For example, the result of the following statement is not in the form $\mathbf{a} + \mathbf{bi}$ because **a** is not a number.

```
1 / ( a - I );      returns   $\frac{1}{a-I}$ 
```

To use the evalc command (the symbolic evaluator over the complex field), we enter

```
evalc ( 1 / ( a - I ) );
```

Question 12 *What is the Maple output from the last command?*

Note: the evalc command assumes that the symbol **a** is *real*. See ? evalc for more information.

1.4 Assign

We can assign *any* MAPLE expression to a name.

```
var := x;      or      term := x * y;
```

MAPLE names can include any alphanumeric characters and underscores, but they cannot start with a number. Define functions by using *arrow notation* (->). For example,

```
f := x -> 2 * x ^ 2 - 3 * x + 4;
```

Question 13 *Now record the output from the following commands*

```
f;      and      f(x);
```

To recall the function that we have just assigned to f , we need $f(x)$. The assignment operator ($:=$) associates a function name with the function definition. Now try

```
f := x -> x ^ 2;
```

Evaluating f at an argument produces the square of f 's argument.

Question 14 Evaluate the expressions

```
f ( 5 );      and      f ( a + 1 );
```

1.5 Unassign

If we have assigned the constant

```
a := 3;
```

and we wish to clear this value so that a is an arbitrary constant, we enter

```
unassign('a');      now      a;      returns a.
```

Question 15 Record the Maple outputs from the following sequence of commands

```
f(x);      unassign('f ( x )');      f(x);
```

1.6 Restart

To clear all stored information in a MAPLE worksheet, enter `restart`;

2 Graphics

2.1 Two dimensions

Enter

```
f := x -> 7 * sin ( x ) + sin ( 7 * x );
```

For a simple plot of this function

```
plot ( f ( x ) , x = - 10 .. 10 );
```

Question 16 Select *File/Preferences/plotting* and click on *window*. Repeat the previous command. What is different about the output?

Now try

```
plot ( sin ( x ) , x = - 2 * Pi .. 2 * Pi );
```

Question 17 Describe the Maple output you have just generated. Now try

```
plot ( sin ( x ) , x = - 3 * Pi .. 7 * Pi );
```

How has your figure changed?

Clicking any point in the plot window shows those particular coordinates of the plot. You can focus on a particular section in the x - and y -dimensions. For example

```
plot ( f ( x ) , x = - 10 .. 10 , y = 4 .. 8 );
```

You can also plot infinite domains

```
plot ( sin ( x ) / x , x = 0 .. infinity );
```

2.1.1 Parametric plots

```
plot ( [ cos ( t ) , sin ( t ) , t = 0 .. 2 * Pi ] );
```

Although the plot should look like a circle, your plot may resemble an ellipse because MAPLE scales the plot to fit the window. You can change this by plotting

```
plot( [ cos ( t ) , sin ( t ) , t = 0 .. 2 * Pi ] , scaling = constrained );
```

2.1.2 Polar coordinates

MAPLE can plot functions in polar coordinates using the `polarplot` command. First you must enter

```
with ( plots ) :
```

Question 18 Try the input

```
with(plots);
```

What is the effect of replacing the `;` at the end of expression with a `?`?

Now to plot a circle of radius 1

```
polarplot ( 1 , theta = 0 .. 2 * Pi , scaling = constrained );
```

Question 19 Try the following four polarplot commands. Sketch the figures. (Alternatively, see if you print them out on the printer!)

```
polarplot ( sin ( 3 * theta ) , theta = 0 .. 2 * Pi );
polarplot ( theta , theta = 0 .. 4 * Pi );
polarplot ( sin ( 3 * y ) , y = 0 .. 2 * Pi );
polarplot ( t , t = 0 .. 4 * Pi );
```

2.1.3 Multiple functions

To graph more than one function in the same plot

```
plot ( [ x , x ^ 2 , x ^ 3 ] , x = -10 .. 10 , y = - 10 .. 10 );
f := x -> sin ( x );
g := x -> cos ( x );
plot ( [ f ( x ) , g ( x ) ] , x = 0 .. 2 * Pi );
```

2.2 Three dimensions

To visualise a function of two variables, for example, $\sin(xy)$,

```
plot3d ( sin ( x * y ) , x = - 2 .. 2 , y = - 2 .. 2 );
```

Question 20 Click your image and move your mouse. Explain what happens to the image.

There are many options available for graphing. See ?plot,options for more information.

3 Differentiation

3.1 Single variable functions

Question 21 Record the output from the following sequence of commands.

```
f := x -> x ^ 4;
diff ( f ( x ) , x );      Diff ( f ( x ) , x );      diff ( f , x );
diff ( sin ( x ) , x );    Diff ( sin ( x ) , x );      value ( % );
diff ( k ( x ) , x );      that is, k is some undefined function of x.
```

For higher order derivatives

```
diff ( f ( x ) , x , x );      diff ( f ( x ) , x $ 2 );      diff ( g ( x ) , x $ 3 );
```

3.2 Functions of several variables

```
h := ( x , y ) -> x ^ 2 + y ^ 2 + x * y ;
```

To find $\frac{\partial h}{\partial x}$, $\frac{\partial h}{\partial y}$ and $\frac{\partial^2 h}{\partial x \partial y}$ enter

```
diff ( h ( x , y ) , x );      diff ( h ( x , y ) , y );      diff ( h ( x , y ) , x , y );
```

4 Integration

Consider $k(x) = \sin(2x)$. Enter

```
k := x -> sin ( 2 * x ) ;
```

Question 22 Record the output from the following commands

```
int ( k ( x ) , x );      Int ( k ( x ) , x );      value ( % );
```

Note MAPLE does not add a constant in indefinite integration.

Question 23 Now try.

```
int ( k ( x ) , x = 1 .. 3 );      Int ( k ( x ) , x = 1 .. 3 );      value ( % );
```

5 Substitution

Define

```
a := x -> x ^ 2;      and      b := x ^ 2;
```

Question 24 Record the output from the following sequence of commands

```
a ;      a ( x ) ;      b ;      b ( x ) ;
```

To evaluate at $x = 3$, try the following

```
a ( 3 ) ;      b ( 3 ) ;
```

Because the functions have been defined differently, we cannot get the same output. To evaluate b at $x = 3$, can use

```
subs ( x = 3 , b ) ;
```

Let $c = 2x + \cos(y) + d1 - 2d2$ by entering

```
c := ( x , y ) -> 2 * x + cos ( y ) + d1 - 2 * d2 ;
```

Note here x and y are independent variables and d1 and d2 are arbitrary constants.

Question 25 Record the output from the following command

```
subs ( x = 3 , y = 2 , d1 = 0 , c ( x , y ) );
```

See ?subs for more information.

Acknowledgements

Original MAPLE worksheet written and devised by Maureen Edwards. Minor changes by Jim Whitehead Mark Nelson.